# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=webenum
pkgver=21.24b43b4
pkgrel=3
pkgdesc='Tool to enumerate http responses using dynamically generated queries and more.'
groups=('blackarch' 'blackarch-scanner' 'blackarch-webapp')
url='https://github.com/sarthakpandit/webenum'
license=('GPL')
arch=('any')
depends=('python2')
makedepends=('git')
source=("git+https://github.com/sarthakpandit/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

prepare() {
  cd $pkgname

  sed -i 's|env python|env python2|' "$pkgname.py"
}

package() {
  cd $pkgname

  install -Dm 755 "$pkgname.py" "$pkgdir/usr/bin/$pkgname"
}

